

# SQL\_SELECT

## SQL\_SELECT action

### Function

The action executes the SQL command SELECT.

### Declaration

```
SQL_SELECT dbObjIdent, retCodeIdent_Int, [maxRowsIdent_Int],
selectIdent_String [TRANS transHandle_Int] BIND ...
```

or

```
SQL_SELECT connectString, retCodeIdent_Int, [maxRowsIdent_Int],
selectIdent_String ON dbManIdent BIND ...
```

The keyword BIND must be followed by one of these parameters:

BIND \_locVar1, \_locVar2, ...

BIND \_locVarRowIdent

BIND \_locVarRecordIdent

### Parameters

|                         |     |                                                                                                                    |
|-------------------------|-----|--------------------------------------------------------------------------------------------------------------------|
| dbObjIdent              | in  | <a href="#">Reference to an object</a> of the <a href="#">Database</a> or <a href="#">Database table</a> type.     |
| connectString           | in  | <a href="#">Identifier</a> of the <i>Text</i> type or a text constant containing a connect string to the database. |
| retCodeIdent_Int        | out | <a href="#">Identifier</a> of the <i>Int</i> type - return code.                                                   |
| maxRowsIdent_Int        | in  | Identifier of the <i>Int</i> type - number of rows read en bloc (if it is not entered the default value is 1).     |
| selectIdent_String      | in  | Identifier of the <i>String</i> type.                                                                              |
| _locVar1, _locVar2, ... | in  | List of local variables.                                                                                           |
| _locVarRowIdent         | in  | <a href="#">Reference</a> to a row of <a href="#">local variable</a> of the <i>Record</i> type.                    |
| _locVarRecordIdent      | in  | <a href="#">Identifier</a> of <a href="#">local variable</a> of the <i>Record</i> type.                            |

### Return code

The value of the parameter *retCodeIdent\_Int* - see the table of [error codes](#). It is possible to get [extended error information](#).

### Description

The action SQL\_SELECT enables to make SQL command SELECT as simple as possible. It combines the actions [SQL\\_CONNECT](#), [SQL\\_PREPARE](#), [SQL\\_FETCH](#) and [SQL\\_DISCONNECT](#).

A database, in which the SQL command SELECT will be executed, can be defined by:

1. parameter *dbObjIdent* - represents the object of *Database* or *Table* type. User can enter the parameter (optional) *transHandle\_Int* that defines a database connection (the value of parameter has been gained by previous calling the action DB\_TRANS\_OPEN), within it the SQL command will be executed, If the parameter *transHandle\_Int* is not defined, SQL command will be executed on one of the predefined [database connections](#).
2. parameter *connectString* (described on page [SQL\\_CONNECT action](#)). It is necessary to specify the process DB Manager that will execute the SQL command.

The command SELECT is defined by value of parameter *selectIdent\_String*. According to the number of columns of SELECT command should be chosen the proper variant to show its result through the obligatory keyword BIND. The meaning and explanation of the variants are mentioned in action [SQL\\_PREPARE](#).

An optional identifier *maxRowsIdent\_Int* defines the row counts of select which will be returned. If it is not defined its value is 1. The number of the rows that can be read depends also on the variant BIND. Details are in the description of action [SQL\\_FETCH](#).

The maximum number of returned rows is limited by a Database configuration parameter [Maximum returned rows](#). If the number of available rows exceeds this limit, the local structure *\_locVarRecordIdent* will contain first [Maximum returned rows](#) and *retCodeIdent\_Int* will be set to error code [\\_ERR\\_DATABASE\\_ROWS\\_LIMIT](#).

## Note

The values returned by the SQL statement are inserted into local variables according to their order (1st value to the first bound variable, 2nd to the second, etc.). This also applies when inserting into a local structured variable (i.e. column names are not taken into account). Potentially dangerous are SELECTs of the "SELECT \*" type - if the order of the columns in the database differs from the order of the columns in the structured variable, the values will be inserted in the wrong columns. Therefore, we recommend explicitly naming columns in SQL query (SELECT COL1, COL2, COL3 ...).

If retrieving data from a single table or view, use [DBS\\_READ](#) instead of SQL\_SELECT.

## Example

In this example, there exists the object of *Database* type with the name **gnat\_test**. There is a table OBJLIST within database in which the column *id* is figure.  
A procedure Demo1 represents various (not all) variants of SQL\_SELECT action.  
The procedures Demo2 and Demo3 have the same function. Demo2 is implemented through the SQL\_SELECT action and Demo3 is implemented by SQL\_CONNECT action, SQL\_PREPARE action, SQL\_FETCH action and SQL\_DISCONNECT action.

```
PROCEDURE Demo1

  INT _retCod
  INT _maxId

  ; select through the object of Database type
  SQL_SELECT gnat_test, _retCode,, "select max(id) from OBJLIST" BIND
  _maxId
  IF _retCode = _ERR_NO_ERROR THEN
    ; variable _maxId contains the value
  ELSE
    ; error
  ENDIF

  ; select through the object of Table type
  SQL_SELECT DB.OBJLIST, _retCode,, "select max(id) from OBJLIST" BIND
  _maxId
  IF _retCode = _ERR_NO_ERROR THEN
    ; variable _maxId contains the value
  ELSE
    ; error
  ENDIF

  INT _transHandle
  ; opening of the transaction through the object of Database type
  DB_TRANS_OPEN gnat_test, _transHandle, _retCode
  IF _retCode = _ERR_NO_ERROR THEN
    ; transaction has been opened
    ; select within the transaction
    SQL_SELECT DB.OBJLIST, _retCode,, "select max(id) from OBJLIST" TRANS
    _transHandle BIND _maxId
    IF _retCode = _ERR_NO_ERROR THEN
      ; variable _maxId contains the value
    ELSE
      ; error
    ENDIF

    DB_TRANS_CLOSE _transHandle
  ELSE
    ; error
  ENDIF

  ; select with the connection
  SQL_SELECT "connection String", _retCode,, "select max(id) from OBJLIST"
  ON SELF.DBM BIND _maxId
  IF _retCode = _ERR_NO_ERROR THEN
    ; variable _maxId contains the value
  ELSE
    ; error
  ENDIF

END Demo1

PROCEDURE Demo2(INT _maxId)
```

```

INT _retCode

_maxId := %SetInvalid(0)
; select through the object of Database type
SQL_SELECT gnat_test, _retCode,, "select max(id) from OBJLIST" BIND
_maxId
IF _retCode = _ERR_NO_ERROR THEN
; variable _maxId contains the value
ELSE
; error
ENDIF

END Demo2

PROCEDURE Demo3(INT _maxId)
INT _retCode
INT _handle
_maxId := %SetInvalid(0)
;
SQL_CONNECT gnat_test, _handle, _retCode
IF _retCode <> _ERR_NO_ERROR THEN
RETURN ; error
ENDIF

SQL_PREPARE _handle, _retCode, "select max(id) from OBJLIST" BIND _maxId
IF _retCode <> _ERR_NO_ERROR THEN
RETURN ; error
ENDIF

SQL_FETCH _handle, _retCode
IF _retCode <> _ERR_NO_ERROR THEN
RETURN ; error
ENDIF

SQL_DISCONNECT _handle

END Demo3

BEGIN

CALL Demo1

INT _max
CALL Demo2(_max)

CALL Demo3(_max)
END

```

#### Related topics

[DB\\_TRANS\\_OPEN](#)  
[DB\\_TRANS\\_COMMIT](#)  
[DB\\_TRANS\\_ROLLBACK](#)  
[DB\\_TRANS\\_CLOSE](#)

[SQL\\_CONNECT](#)  
[SQL\\_EXEC\\_DIRECT](#)  
[SQL\\_EXEC\\_PROC](#)

[SQL\\_PREPARE](#)  
[SQL\\_BINDIN](#)  
[SQL\\_FETCH](#)  
[SQL\\_FREE](#)

[Actions for accessing a database.](#)



**Related pages:**

[Script actions](#)